// File: @openzeppelin/contracts/utils/Base64.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (utils/Base64.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides a set of functions to operate with Base64 strings.
 *
 * _Available since v4.5._
 */
library Base64 {
    /**
     * @dev Base64 Encoding/Decoding Table
     */
    string internal constant _TABLE = "ABCDEFGHIJKLMNOPQRSTUVWXYZabcdefghijklmnopqrstuvwxyz0123456789+/";

    /**
     * @dev Converts a `bytes` to its Bytes64 `string` representation.
     */
    function encode(bytes memory data) internal pure returns (string memory) {
        /**
         * Inspired by Brecht Devos (Brechtpd) implementation - MIT licence
         * https://github.com/Brechtpd/base64/blob/e78d9fd951e7b0977ddca77d92dc85183770daf4/base64.sol
         */
        if (data.length == 0) return "";

        // Loads the table into memory
        string memory table = _TABLE;

        // Encoding takes 3 bytes chunks of binary data from `bytes` data parameter
        // and split into 4 numbers of 6 bits.
        // The final Base64 length should be `bytes` data length multiplied by 4/3 rounded up
        // - `data.length + 2`  -> Round up
        // - `/ 3`              -> Number of 3-bytes chunks
        // - `4 *`              -> 4 characters for each chunk
        string memory result = new string(4 * ((data.length + 2) / 3));

        /// @solidity memory-safe-assembly
        assembly {
            // Prepare the lookup table (skip the first "length" byte)
            let tablePtr := add(table, 1)

            // Prepare result pointer, jump over length
            let resultPtr := add(result, 32)

            // Run over the input, 3 bytes at a time
            for {
                let dataPtr := data
                let endPtr := add(data, mload(data))
            } lt(dataPtr, endPtr) {

            } {
                // Advance 3 bytes
                dataPtr := add(dataPtr, 3)
                let input := mload(dataPtr)

                // To write each character, shift the 3 bytes (18 bits) chunk
                // 4 times in blocks of 6 bits for each character (18, 12, 6, 0)
                // and apply logical AND with 0x3F which is the number of
                // the previous character in the ASCII table prior to the Base64 Table
                // The result is then added to the table to get the character to write,
                // and finally write it in the result pointer but with a left shift
                // of 256 (1 byte) - 8 (1 ASCII char) = 248 bits

                mstore8(resultPtr, mload(add(tablePtr, and(shr(18, input), 0x3F))))
                resultPtr := add(resultPtr, 1) // Advance

                mstore8(resultPtr, mload(add(tablePtr, and(shr(12, input), 0x3F))))
                resultPtr := add(resultPtr, 1) // Advance

                mstore8(resultPtr, mload(add(tablePtr, and(shr(6, input), 0x3F))))
                resultPtr := add(resultPtr, 1) // Advance

                mstore8(resultPtr, mload(add(tablePtr, and(input, 0x3F))))
                resultPtr := add(resultPtr, 1) // Advance
            }

            // When data `bytes` is not exactly 3 bytes long
            // it is padded with `=` characters at the end
            switch mod(mload(data), 3)
            case 1 {
                mstore8(sub(resultPtr, 1), 0x3d)
                mstore8(sub(resultPtr, 2), 0x3d)
            }
            case 2 {
                mstore8(sub(resultPtr, 1), 0x3d)
            }
        }

        return result;
    }
}

// File: @openzeppelin/contracts/utils/math/Math.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (utils/math/Math.sol)

pragma solidity ^0.8.0;

/**
 * @dev Standard math utilities missing in the Solidity language.
 */
library Math {
    enum Rounding {
        Down, // Toward negative infinity
        Up, // Toward infinity
        Zero // Toward zero
    }

    /**
     * @dev Returns the largest of two numbers.
     */
    function max(uint256 a, uint256 b) internal pure returns (uint256) {
        return a > b ? a : b;
    }

    /**
     * @dev Returns the smallest of two numbers.
     */
    function min(uint256 a, uint256 b) internal pure returns (uint256) {
        return a < b ? a : b;
    }

    /**
     * @dev Returns the average of two numbers. The result is rounded towards
     * zero.
     */
    function average(uint256 a, uint256 b) internal pure returns (uint256) {
        // (a + b) / 2 can overflow.
        return (a & b) + (a ^ b) / 2;
    }

    /**
     * @dev Returns the ceiling of the division of two numbers.
     *
     * This differs from standard division with `/` in that it rounds up instead
     * of rounding down.
     */
    function ceilDiv(uint256 a, uint256 b) internal pure returns (uint256) {
        // (a + b - 1) / b can overflow on addition, so we distribute.
        return a == 0 ? 0 : (a - 1) / b + 1;
    }

    /**
     * @notice Calculates floor(x * y / denominator) with full precision. Throws if result overflows a uint256 or denominator == 0
     * @dev Original credit to Remco Bloemen under MIT license (https://xn--2-umb.com/21/muldiv)
     * with further edits by Uniswap Labs also under MIT license.
     */
    function mulDiv(uint256 x, uint256 y, uint256 denominator) internal pure returns (uint256 result) {
        unchecked {
            // 512-bit multiply [prod1 prod0] = x * y. Compute the product mod 2^256 and mod 2^256 - 1, then use
            // use the Chinese Remainder Theorem to reconstruct the 512 bit result. The result is stored in two 256
            // variables such that product = prod1 * 2^256 + prod0.
            uint256 prod0; // Least significant 256 bits of the product
            uint256 prod1; // Most significant 256 bits of the product
            assembly {
                let mm := mulmod(x, y, not(0))
                prod0 := mul(x, y)
                prod1 := sub(sub(mm, prod0), lt(mm, prod0))
            }

            // Handle non-overflow cases, 256 by 256 division.
            if (prod1 == 0) {
                // Solidity will revert if denominator == 0, unlike the div opcode on its own.
                // The surrounding unchecked block does not change this fact.
                // See https://docs.soliditylang.org/en/latest/control-structures.html#checked-or-unchecked-arithmetic.
                return prod0 / denominator;
            }

            // Make sure the result is less than 2^256. Also prevents denominator == 0.
            require(denominator > prod1, "Math: mulDiv overflow");

            ///////////////////////////////////////////////
            // 512 by 256 division.
            ///////////////////////////////////////////////

            // Make division exact by subtracting the remainder from [prod1 prod0].
            uint256 remainder;
            assembly {
                // Compute remainder using mulmod.
                remainder := mulmod(x, y, denominator)

                // Subtract 256 bit number from 512 bit number.
                prod1 := sub(prod1, gt(remainder, prod0))
                prod0 := sub(prod0, remainder)
            }

            // Factor powers of two out of denominator and compute largest power of two divisor of denominator. Always >= 1.
            // See https://cs.stackexchange.com/q/138556/92363.

            // Does not overflow because the denominator cannot be zero at this stage in the function.
            uint256 twos = denominator & (~denominator + 1);
            assembly {
                // Divide denominator by twos.
                denominator := div(denominator, twos)

                // Divide [prod1 prod0] by twos.
                prod0 := div(prod0, twos)

                // Flip twos such that it is 2^256 / twos. If twos is zero, then it becomes one.
                twos := add(div(sub(0, twos), twos), 1)
            }

            // Shift in bits from prod1 into prod0.
            prod0 |= prod1 * twos;

            // Invert denominator mod 2^256. Now that denominator is an odd number, it has an inverse modulo 2^256 such
            // that denominator * inv = 1 mod 2^256. Compute the inverse by starting with a seed that is correct for
            // four bits. That is, denominator * inv = 1 mod 2^4.
            uint256 inverse = (3 * denominator) ^ 2;

            // Use the Newton-Raphson iteration to improve the precision. Thanks to Hensel's lifting lemma, this also works
            // in modular arithmetic, doubling the correct bits in each step.
            inverse *= 2 - denominator * inverse; // inverse mod 2^8
            inverse *= 2 - denominator * inverse; // inverse mod 2^16
            inverse *= 2 - denominator * inverse; // inverse mod 2^32
            inverse *= 2 - denominator * inverse; // inverse mod 2^64
            inverse *= 2 - denominator * inverse; // inverse mod 2^128
            inverse *= 2 - denominator * inverse; // inverse mod 2^256

            // Because the division is now exact we can divide by multiplying with the modular inverse of denominator.
            // This will give us the correct result modulo 2^256. Since the preconditions guarantee that the outcome is
            // less than 2^256, this is the final result. We don't need to compute the high bits of the result and prod1
            // is no longer required.
            result = prod0 * inverse;
            return result;
        }
    }

    /**
     * @notice Calculates x * y / denominator with full precision, following the selected rounding direction.
     */
    function mulDiv(uint256 x, uint256 y, uint256 denominator, Rounding rounding) internal pure returns (uint256) {
        uint256 result = mulDiv(x, y, denominator);
        if (rounding == Rounding.Up && mulmod(x, y, denominator) > 0) {
            result += 1;
        }
        return result;
    }

    /**
     * @dev Returns the square root of a number. If the number is not a perfect square, the value is rounded down.
     *
     * Inspired by Henry S. Warren, Jr.'s "Hacker's Delight" (Chapter 11).
     */
    function sqrt(uint256 a) internal pure returns (uint256) {
        if (a == 0) {
            return 0;
        }

        // For our first guess, we get the biggest power of 2 which is smaller than the square root of the target.
        //
        // We know that the "msb" (most significant bit) of our target number `a` is a power of 2 such that we have
        // `msb(a) <= a < 2*msb(a)`. This value can be written `msb(a)=2**k` with `k=log2(a)`.
        //
        // This can be rewritten `2**log2(a) <= a < 2**(log2(a) + 1)`
        // → `sqrt(2**k) <= sqrt(a) < sqrt(2**(k+1))`
        // → `2**(k/2) <= sqrt(a) < 2**((k+1)/2) <= 2**(k/2 + 1)`
        //
        // Consequently, `2**(log2(a) / 2)` is a good first approximation of `sqrt(a)` with at least 1 correct bit.
        uint256 result = 1 << (log2(a) >> 1);

        // At this point `result` is an estimation with one bit of precision. We know the true value is a uint128,
        // since it is the square root of a uint256. Newton's method converges quadratically (precision doubles at
        // every iteration). We thus need at most 7 iteration to turn our partial result with one bit of precision
        // into the expected uint128 result.
        unchecked {
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            return min(result, a / result);
        }
    }

    /**
     * @notice Calculates sqrt(a), following the selected rounding direction.
     */
    function sqrt(uint256 a, Rounding rounding) internal pure returns (uint256) {
        unchecked {
            uint256 result = sqrt(a);
            return result + (rounding == Rounding.Up && result * result < a ? 1 : 0);
        }
    }

    /**
     * @dev Return the log in base 2, rounded down, of a positive value.
     * Returns 0 if given 0.
     */
    function log2(uint256 value) internal pure returns (uint256) {
        uint256 result = 0;
        unchecked {
            if (value >> 128 > 0) {
                value >>= 128;
                result += 128;
            }
            if (value >> 64 > 0) {
                value >>= 64;
                result += 64;
            }
            if (value >> 32 > 0) {
                value >>= 32;
                result += 32;
            }
            if (value >> 16 > 0) {
                value >>= 16;
                result += 16;
            }
            if (value >> 8 > 0) {
                value >>= 8;
                result += 8;
            }
            if (value >> 4 > 0) {
                value >>= 4;
                result += 4;
            }
            if (value >> 2 > 0) {
                value >>= 2;
                result += 2;
            }
            if (value >> 1 > 0) {
                result += 1;
            }
        }
        return result;
    }

    /**
     * @dev Return the log in base 2, following the selected rounding direction, of a positive value.
     * Returns 0 if given 0.
     */
    function log2(uint256 value, Rounding rounding) internal pure returns (uint256) {
        unchecked {
            uint256 result = log2(value);
            return result + (rounding == Rounding.Up && 1 << result < value ? 1 : 0);
        }
    }

    /**
     * @dev Return the log in base 10, rounded down, of a positive value.
     * Returns 0 if given 0.
     */
    function log10(uint256 value) internal pure returns (uint256) {
        uint256 result = 0;
        unchecked {
            if (value >= 10 ** 64) {
                value /= 10 ** 64;
                result += 64;
            }
            if (value >= 10 ** 32) {
                value /= 10 ** 32;
                result += 32;
            }
            if (value >= 10 ** 16) {
                value /= 10 ** 16;
                result += 16;
            }
            if (value >= 10 ** 8) {
                value /= 10 ** 8;
                result += 8;
            }
            if (value >= 10 ** 4) {
                value /= 10 ** 4;
                result += 4;
            }
            if (value >= 10 ** 2) {
                value /= 10 ** 2;
                result += 2;
            }
            if (value >= 10 ** 1) {
                result += 1;
            }
        }
        return result;
    }

    /**
     * @dev Return the log in base 10, following the selected rounding direction, of a positive value.
     * Returns 0 if given 0.
     */
    function log10(uint256 value, Rounding rounding) internal pure returns (uint256) {
        unchecked {
            uint256 result = log10(value);
            return result + (rounding == Rounding.Up && 10 ** result < value ? 1 : 0);
        }
    }

    /**
     * @dev Return the log in base 256, rounded down, of a positive value.
     * Returns 0 if given 0.
     *
     * Adding one to the result gives the number of pairs of hex symbols needed to represent `value` as a hex string.
     */
    function log256(uint256 value) internal pure returns (uint256) {
        uint256 result = 0;
        unchecked {
            if (value >> 128 > 0) {
                value >>= 128;
                result += 16;
            }
            if (value >> 64 > 0) {
                value >>= 64;
                result += 8;
            }
            if (value >> 32 > 0) {
                value >>= 32;
                result += 4;
            }
            if (value >> 16 > 0) {
                value >>= 16;
                result += 2;
            }
            if (value >> 8 > 0) {
                result += 1;
            }
        }
        return result;
    }

    /**
     * @dev Return the log in base 256, following the selected rounding direction, of a positive value.
     * Returns 0 if given 0.
     */
    function log256(uint256 value, Rounding rounding) internal pure returns (uint256) {
        unchecked {
            uint256 result = log256(value);
            return result + (rounding == Rounding.Up && 1 << (result << 3) < value ? 1 : 0);
        }
    }
}

// File: @openzeppelin/contracts/utils/math/SignedMath.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (utils/math/SignedMath.sol)

pragma solidity ^0.8.0;

/**
 * @dev Standard signed math utilities missing in the Solidity language.
 */
library SignedMath {
    /**
     * @dev Returns the largest of two signed numbers.
     */
    function max(int256 a, int256 b) internal pure returns (int256) {
        return a > b ? a : b;
    }

    /**
     * @dev Returns the smallest of two signed numbers.
     */
    function min(int256 a, int256 b) internal pure returns (int256) {
        return a < b ? a : b;
    }

    /**
     * @dev Returns the average of two signed numbers without overflow.
     * The result is rounded towards zero.
     */
    function average(int256 a, int256 b) internal pure returns (int256) {
        // Formula from the book "Hacker's Delight"
        int256 x = (a & b) + ((a ^ b) >> 1);
        return x + (int256(uint256(x) >> 255) & (a ^ b));
    }

    /**
     * @dev Returns the absolute unsigned value of a signed value.
     */
    function abs(int256 n) internal pure returns (uint256) {
        unchecked {
            // must be unchecked in order to support `n = type(int256).min`
            return uint256(n >= 0 ? n : -n);
        }
    }
}

// File: @openzeppelin/contracts/utils/Strings.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (utils/Strings.sol)

pragma solidity ^0.8.0;

import "./math/Math.sol";
import "./math/SignedMath.sol";

/**
 * @dev String operations.
 */
library Strings {
    bytes16 private constant _SYMBOLS = "0123456789abcdef";
    uint8 private constant _ADDRESS_LENGTH = 20;

    /**
     * @dev Converts a `uint256` to its ASCII `string` decimal representation.
     */
    function toString(uint256 value) internal pure returns (string memory) {
        unchecked {
            uint256 length = Math.log10(value) + 1;
            string memory buffer = new string(length);
            uint256 ptr;
            /// @solidity memory-safe-assembly
            assembly {
                ptr := add(buffer, add(32, length))
            }
            while (true) {
                ptr--;
                /// @solidity memory-safe-assembly
                assembly {
                    mstore8(ptr, byte(mod(value, 10), _SYMBOLS))
                }
                value /= 10;
                if (value == 0) break;
            }
            return buffer;
        }
    }

    /**
     * @dev Converts a `int256` to its ASCII `string` decimal representation.
     */
    function toString(int256 value) internal pure returns (string memory) {
        return string(abi.encodePacked(value < 0 ? "-" : "", toString(SignedMath.abs(value))));
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation.
     */
    function toHexString(uint256 value) internal pure returns (string memory) {
        unchecked {
            return toHexString(value, Math.log256(value) + 1);
        }
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation with fixed length.
     */
    function toHexString(uint256 value, uint256 length) internal pure returns (string memory) {
        bytes memory buffer = new bytes(2 * length + 2);
        buffer[0] = "0";
        buffer[1] = "x";
        for (uint256 i = 2 * length + 1; i > 1; --i) {
            buffer[i] = _SYMBOLS[value & 0xf];
            value >>= 4;
        }
        require(value == 0, "Strings: hex length insufficient");
        return string(buffer);
    }

    /**
     * @dev Converts an `address` with fixed length of 20 bytes to its not checksummed ASCII `string` hexadecimal representation.
     */
    function toHexString(address addr) internal pure returns (string memory) {
        return toHexString(uint256(uint160(addr)), _ADDRESS_LENGTH);
    }

    /**
     * @dev Returns true if the two strings are equal.
     */
    function equal(string memory a, string memory b) internal pure returns (bool) {
        return keccak256(bytes(a)) == keccak256(bytes(b));
    }
}

// File: contracts/erc721/custom/[untitled]/CuratedRendererV1.sol



//SPDX-License-Identifier: UNLICENSED
pragma solidity 0.8.10;

import "../../inchain-rendering/interfaces/IHLRenderer.sol";
import "../../interfaces/IERC721GeneralSupplyMetadata.sol";
import "./interfaces/IHLFS.sol";
import "@openzeppelin/contracts/utils/Strings.sol";
import "@openzeppelin/contracts/utils/Base64.sol";
import "./interfaces/IOwnable.sol";

/**
 * @notice HL in-chain Renderer for curated hash based projects
 * @dev Currently supports a fairly fixed configuratoin
 * @author highlight.xyz
 */
contract CuratedRendererV1 {
    /**
     * @notice Throw when invalid input to process mint data
     */
    error InvalidMintData();

    /**
     * @notice Throw when seed input for a token cannot be found
     */
    error SeedInputNotFound();

    /**
     * @notice Throw when transaction sender isn't collection owner
     */
    error NotCollectionOwner();

    /**
     * @notice Input that seeds token metadata
     * @param previousBlockHash Hash of the block before the one the tokens were minted on
     * @param blockTimestamp Timestamp of block that tokens were minted on
     * @param startTokenId ID of first token of minted batch
     */
    struct SeedInput {
        bytes32 previousBlockHash;
        uint48 blockTimestamp;
        uint176 startTokenId;
        uint32 numMinted;
    }

    /**
     * @notice Simple collection config (to be made more complex)
     */
    struct CollectionConfig {
        string name;
        string previewsBaseUri;
        string htmlLang;
        bool htmlBodyExpected;
        bool useCDN;
    }

    uint256 private constant _32_BIT_MAX_MINUS_1 = 2 ** 32 - 1;

    /**
     * @notice Store the seed inputs for each token batch (for each nft contract)
     * @dev Assume startTokenIds are incrementing (up to implementer), assume first batch's startTokenId is 1
     */
    mapping(address => SeedInput[]) public collectionSeedInputs;

    /**
     * @notice Store config per collection
     */
    mapping(address => CollectionConfig) public collectionConfig;

    /**
     * @notice Set a collection's config
     */
    function setCollectionConfig(CollectionConfig calldata config, address collection) external {
        if (IOwnable(collection).owner() == msg.sender || msg.sender == collection) {
            collectionConfig[collection] = config;
        } else {
            _revert(NotCollectionOwner.selector);
        }
    }

    /**
     * @notice See {IHlRenderer-processOneRecipientMint}
     */
    function processOneRecipientMint(uint256 firstTokenId, uint256 numTokens, address recipient) external {
        collectionSeedInputs[msg.sender].push(
            SeedInput(blockhash(block.number - 1), uint48(block.timestamp), uint176(firstTokenId), uint32(numTokens))
        );
    }

    /**
     * @notice See {IHLRenderer-tokenURI}
     */
    function tokenURI(uint256 tokenId) public view virtual returns (string memory) {
        address collection = msg.sender;
        SeedInput memory _seedInput = getSeedInput(tokenId, collection);
        bytes32 curatedHash = getCuratedHash(tokenId, collection);

        string memory tokenIdStr = Strings.toString(tokenId);

        bytes[] memory metadata = new bytes[](3);
        bytes[] memory encodedJson = new bytes[](2);

        // inject values in JS
        string memory injectedToken = string(
            abi.encodePacked(
                'const injectedToken = {"blockHash": "',
                Strings.toHexString(uint256(_seedInput.previousBlockHash)),
                '", "',
                'tokenId": "',
                tokenIdStr,
                '", "',
                'timestamp": "',
                Strings.toString(_seedInput.blockTimestamp),
                '", "',
                'hash": "',
                Strings.toHexString(uint256(curatedHash)),
                '", "',
                'isCurated": "1"',
                "};"
            )
        );

        metadata[0] = abi.encodePacked(
            '{"name": "',
            collectionConfig[collection].name,
            " #",
            tokenIdStr,
            '", "',
            'description": "',
            IHLFS(collection).fileContents("description.txt"),
            '", "',
            'image": "',
            collectionConfig[collection].previewsBaseUri,
            "/",
            Strings.toString(tokenId),
            ".png",
            '", "'
            'animation_url": "data:text/html;base64,'
        );
        metadata[1] = bytes(Base64.encode(_generateHTML(collection, injectedToken, false)));
        metadata[2] = bytes('"}');

        encodedJson[0] = bytes("data:application/json;base64,");
        encodedJson[1] = bytes(Base64.encode(concat(metadata)));
        return string(concat(encodedJson));
    }

    /**
     * @notice See {IHLRenderer-tokenURI}
     */
    function tokenURIWithCDN(uint256 tokenId, address collection) public view virtual returns (string memory) {
        SeedInput memory _seedInput = getSeedInput(tokenId, collection);
        bytes32 curatedHash = getCuratedHash(tokenId, collection);

        string memory tokenIdStr = Strings.toString(tokenId);

        bytes[] memory metadata = new bytes[](3);
        bytes[] memory encodedJson = new bytes[](2);

        // inject values in JS
        string memory injectedToken = string(
            abi.encodePacked(
                'const injectedToken = {"blockHash": "',
                Strings.toHexString(uint256(_seedInput.previousBlockHash)),
                '", "',
                'tokenId": "',
                tokenIdStr,
                '", "',
                'timestamp": "',
                Strings.toString(_seedInput.blockTimestamp),
                '", "',
                'hash": "',
                Strings.toHexString(uint256(curatedHash)),
                '", "',
                'isCurated": "1"',
                "};"
            )
        );

        metadata[0] = abi.encodePacked(
            '{"name": "',
            collectionConfig[collection].name,
            " #",
            tokenIdStr,
            '", "',
            'description": "',
            IHLFS(collection).fileContents("description.txt"),
            '", "',
            'image": "',
            collectionConfig[collection].previewsBaseUri,
            "/",
            Strings.toString(tokenId),
            ".png",
            '", "'
            'animation_url": "data:text/html;base64,'
        );
        metadata[1] = bytes(Base64.encode(_generateHTML(collection, injectedToken, true)));
        metadata[2] = bytes('"}');

        encodedJson[0] = bytes("data:application/json;base64,");
        encodedJson[1] = bytes(Base64.encode(concat(metadata)));
        return string(concat(encodedJson));
    }

    /**
     * @notice Get a token's curated hash
     */
    function getCuratedHash(uint256 tokenId, address collection) public view returns (bytes32) {
        bytes32[] memory curatedHashes = _parseCuratedHashText(
            IHLFS(collection).fileContents("curatedHashes.txt"),
            IERC721GeneralSupplyMetadata(collection).limitSupply()
        );
        uint256 initialCuratedHashesLength = curatedHashes.length;
        bytes32 lastCuratedHash = bytes32(0);

        for (uint256 i = 0; i < tokenId; i++) {
            SeedInput memory _seedInput = getSeedInput(i + 1, collection);
            bytes32 seed = _getSeed(_seedInput, i + 1);
            uint256 generatedIndex = _prngSeedInput(
                seed,
                _seedInput.previousBlockHash,
                i + 1,
                initialCuratedHashesLength - i
            );
            uint256 virtualIndexPlusOne = 0;

            for (uint256 j = 0; j < initialCuratedHashesLength; j++) {
                if (curatedHashes[j] != bytes32(0)) {
                    virtualIndexPlusOne += 1;
                    if (virtualIndexPlusOne == generatedIndex + 1) {
                        // curated hash found for final token id
                        if (i == tokenId - 1) {
                            lastCuratedHash = curatedHashes[j];
                        }
                        curatedHashes[j] = bytes32(0);
                        break;
                    }
                }
            }
        }

        return lastCuratedHash;
    }

    /**
     * Get a token's seed
     */
    function getSeed(uint256 tokenId, address collection) public view returns (bytes32) {
        SeedInput memory _seedInput = getSeedInput(tokenId, collection);
        return _getSeed(_seedInput, tokenId);
    }

    /**
     * @notice Get a token's seed input
     */
    function getSeedInput(uint256 tokenId, address collection) public view returns (SeedInput memory) {
        SeedInput[] memory _seedInputs = collectionSeedInputs[collection];
        uint256 numInputs = _seedInputs.length;
        if (numInputs == 0) {
            _revert(SeedInputNotFound.selector);
        }
        for (uint256 i = numInputs - 1; i >= 0; i--) {
            if (tokenId >= _seedInputs[i].startTokenId) {
                // assume first batch's startTokenId is 1
                if (_seedInputs[i].startTokenId + _seedInputs[i].numMinted <= tokenId) {
                    _revert(SeedInputNotFound.selector);
                } else {
                    return _seedInputs[i];
                }
            }
        }
    }

    /**
     * @notice Generate the project's HTML file
     */
    function _generateHTML(address collection, string memory injectedToken, bool useCDNOverride) public view returns (bytes memory) {
        bytes[] memory html = new bytes[](4);

        bool useCDN = useCDNOverride ? true : collectionConfig[collection].useCDN;
        string memory libScriptPrefix = useCDN ? ' src="' : ">";
        string memory libScriptSuffix = useCDN ? '">' : "";
        html[0] = abi.encodePacked(
            '<!DOCTYPE html><html lang="',
            collectionConfig[collection].htmlLang,
            '">',
            "<head>",
            IHLFS(collection).fileContents("headPrefix.html"),
            "<script>",
            injectedToken,
            IHLFS(collection).fileContents("hl-gen-inchain.js"),
            "</script><script",
            libScriptPrefix
        );
        html[1] = bytes(
            IHLFS(collection).fileContents(useCDN ? "p5-cdn.txt" : "p5.min.js")
        );
        html[2] = abi.encodePacked(
            libScriptSuffix,
            "</script><script>",
            IHLFS(collection).fileContents("sketch.js"),
            "</script>",
            "<style>",
            IHLFS(collection).fileContents("index.css"),
            "</style>",
            "</head>"
        );

        string memory HTMLBody = "";
        if (collectionConfig[collection].htmlBodyExpected) {
            HTMLBody = IHLFS(collection).fileContents("body.html");
        }
        html[3] = abi.encodePacked("<body>", HTMLBody, "</body>", "</html>");

        return concat(html);
    }

    /**
     * @notice Base64 Encode a metadata JSON
     */
    function _encodeMetadataJSON(bytes memory json) private pure returns (string memory) {
        return string(abi.encodePacked("data:application/json;base64,", Base64.encode(json)));
    }

    /**
     * @notice Concatenate byte arrays
     */
    function concat(bytes[] memory arrays) public pure returns (bytes memory) {
        uint totalLength = 0;
        for (uint i = 0; i < arrays.length; i++) {
            totalLength += arrays[i].length;
        }

        bytes memory result = new bytes(totalLength);
        uint resultPtr;
        assembly {
            resultPtr := add(result, 0x20)
        }

        for (uint i = 0; i < arrays.length; i++) {
            bytes memory array = arrays[i];
            uint arrayLength = array.length;

            uint arrayPtr;
            assembly {
                arrayPtr := add(array, 0x20)
            }

            // Efficiently copy memory block
            for (uint j = 0; j < arrayLength; j += 32) {
                assembly {
                    let chunk := mload(add(arrayPtr, j))
                    mstore(add(resultPtr, j), chunk)
                }
            }

            resultPtr += arrayLength;
        }

        return result;
    }

    /**
     * @notice Parse out all curated hashes for a mint
     */
    function _parseCuratedHashText(
        string memory curatedHashesText,
        uint256 numLines
    ) private pure returns (bytes32[] memory) {
        bytes32[] memory curatedHashes = new bytes32[](numLines);

        uint256 arrayIndex = 0;
        bytes memory stringBytes = bytes(curatedHashesText);
        uint256 i = 0;

        while (i < stringBytes.length && arrayIndex < numLines) {
            // Skip the "0x" prefix at the start of each line
            if (i == 0 || stringBytes[i - 1] == "\n") {
                i += 2;
            }

            bytes32 line;
            for (uint j = 0; j < 32; j++) {
                // Convert two hex characters to one byte
                bytes1 b1 = _parseHexChar(stringBytes[i]);
                bytes1 b2 = _parseHexChar(stringBytes[i + 1]);
                line |= bytes32((uint8(b1) * 16 + uint8(b2)) * 2 ** (8 * (31 - j)));
                i += 2;
            }

            curatedHashes[arrayIndex] = line;
            arrayIndex++;

            // Skip the newline character
            if (i < stringBytes.length && stringBytes[i] == "\n") {
                i++;
            }
        }

        return curatedHashes;
    }

    /**
     * @notice Parse a byte value represented in string to the byte representation of the value
     */
    function _parseHexChar(bytes1 char) internal pure returns (bytes1) {
        if (uint8(char) >= 48 && uint8(char) <= 57) {
            return bytes1(uint8(char) - 48); // 0-9
        }
        if (uint8(char) >= 65 && uint8(char) <= 70) {
            return bytes1(uint8(char) - 55); // A-F
        }
        if (uint8(char) >= 97 && uint8(char) <= 102) {
            return bytes1(uint8(char) - 87); // a-f
        }
        revert("Invalid hex character");
    }

    /**
     * @notice Generate an index via a prng, given seed input and a max value
     */
    function _prngSeedInput(
        bytes32 generalHash,
        bytes32 blockHash,
        uint256 tokenId,
        uint256 max
    ) private pure returns (uint256) {
        uint256 seed = tokenId;

        // process each byte of generalHash and blockHash
        for (uint256 i = 0; i < 32; i++) {
            // extract and process high and low nibbles for generalHash
            uint8 highNibbleTx = uint8(generalHash[i]) >> 4;
            uint8 lowNibbleTx = uint8(generalHash[i]) & 0x0F;
            if (highNibbleTx <= 9) {
                seed += highNibbleTx;
            }
            if (lowNibbleTx <= 9) {
                seed += lowNibbleTx;
            }

            // extract and process high and low nibbles for blockHash
            uint8 highNibbleBlock = uint8(blockHash[i]) >> 4;
            uint8 lowNibbleBlock = uint8(blockHash[i]) & 0x0F;
            if (highNibbleBlock <= 9) {
                seed += highNibbleBlock;
            }
            if (lowNibbleBlock <= 9) {
                seed += lowNibbleBlock;
            }
        }

        uint256 t = (seed + 0x6D2B79F5) & (_32_BIT_MAX_MINUS_1);
        t = imul(t ^ (t >> 15), t | 1) & (_32_BIT_MAX_MINUS_1);
        t ^= (t + imul(t ^ (t >> 7), t | 61)) & (_32_BIT_MAX_MINUS_1);
        t = (t ^ (t >> 14)) & (_32_BIT_MAX_MINUS_1);
        return t % max;
    }

    /**
     * @notice Replicate js Math.imul
     */
    function imul(uint256 a, uint256 b) private pure returns (uint256) {
        return (a * b) % (_32_BIT_MAX_MINUS_1 + 1);
    }

    /**
     * @notice Get a token's seed util
     */
    function _getSeed(SeedInput memory _seedInput, uint256 tokenId) private view returns (bytes32) {
        return keccak256(abi.encodePacked(_seedInput.previousBlockHash, tokenId, _seedInput.blockTimestamp));
    }

    /**
     * @notice Efficient revert
     */
    function _revert(bytes4 errorSelector) private pure {
        assembly {
            mstore(0x00, errorSelector)
            revert(0x00, 0x04)
        }
    }
}

// File: contracts/erc721/custom/[untitled]/interfaces/IHLFS.sol



//SPDX-License-Identifier: UNLICENSED
pragma solidity 0.8.10;

interface IHLFS {
    /**
     * @notice Get contents of a file on a HL FS
     */
    function fileContents(string calldata fileName) external view returns (string memory);
}

// File: contracts/erc721/custom/[untitled]/interfaces/IOwnable.sol



// SPDX-License-Identifier: Unlicensed
pragma solidity 0.8.10;

/**
 * @notice Simple interface to interact with EIP-173 implementing contracts
 */
interface IOwnable {
    function owner() external view returns (address);
}

// File: contracts/erc721/inchain-rendering/interfaces/IHLRenderer.sol



//SPDX-License-Identifier: UNLICENSED
pragma solidity 0.8.10;

/**
 * @notice Highlight's custom renderer interface for collections
 */
interface IHLRenderer {
    /**
     * @notice Process a mint to multiple recipients (likely store mint details)
     * @dev Implementations should assume msg.sender to be the NFT contract
     * @param firstTokenId ID of first token to be minted (next ones are minted sequentially)
     * @param numTokensPerRecipient Number of tokens minted to each recipient
     * @param orderedRecipients Recipients to mint tokens to, sequentially
     */
    function processMultipleRecipientMint(
        uint256 firstTokenId,
        uint256 numTokensPerRecipient,
        address[] calldata orderedRecipients
    ) external;

    /**
     * @notice Process a mint to one recipient (likely store mint details)
     * @dev Implementations should assume msg.sender to be the NFT contract
     * @param firstTokenId ID of first token to be minted (next ones are minted sequentially)
     * @param numTokens Number of tokens minted
     * @param recipient Recipient to mint to
     */
    function processOneRecipientMint(uint256 firstTokenId, uint256 numTokens, address recipient) external;

    /**
     * @notice Return token metadata for a token
     * @dev Implementations should assume msg.sender to be the NFT contract
     * @param tokenId ID of token to return metadata for
     */
    function tokenURI(uint256 tokenId) external view returns (string memory);
}

// File: contracts/erc721/interfaces/IERC721GeneralSupplyMetadata.sol



// SPDX-License-Identifier: UNLICENSED
pragma solidity 0.8.10;

/**
 * @notice Get a Series based collection's supply metadata
 * @author highlight.xyz
 */
interface IERC721GeneralSupplyMetadata {
    /**
     * @notice Get a series based collection's supply, burned tokens notwithstanding
     */
    function supply() external view returns (uint256);

    /**
     * @notice Get a series based collection's total supply
     */
    function totalSupply() external view returns (uint256);

    /**
     * @notice Get a series based collection's supply cap
     */
    function limitSupply() external view returns (uint256);
}